Conjugacy class name of the product in ATLAS Well, I'm trying to read "ATLAS of Finite Groups". To be more precise, I'm interested in character tables of some Weyl groups.
Is it possible to determine the conjugacy class name of the product of two elements from given two classes? For example, if $x$ belongs to 2A and $y$ belongs to $3B$ then what class $xy$ belongs to?
Thanks!
 A: Derek Holt's comment should answer the question, but I find such calculations to be a hassle to do by hand.  It is easier to ask a computer to compute them for you using GAP and its Character Table Library.

describeCCMult := function( ct, A, B )
  local ret, C, k;
  ret := [];
  if not IsPosInt(A) then A := Position( ClassNames( ct ), LowercaseString(A) ); fi;
  if not IsPosInt(B) then B := Position( ClassNames( ct ), LowercaseString(B) ); fi;
  for C in [1..NrConjugacyClasses(ct)] do
    k := ClassMultiplicationCoefficient( ct, A, B, C );
    if k = 1 then Add( ret, UppercaseString(ClassNames(ct))[C] );
    elif k > 0 then Add( ret, Concatenation( String(k), "*", UppercaseString(ClassNames(ct)[C]) ) );
    fi;
  od;
  return JoinStringsWithSeparator( ret, " + " );
end;
ct := CharacterTable("WeylB",6); # the Weyl group of type B6
describeCCMult(ct,"2A", "2B" );

gives output

"5*2A + 3*2C"

meaning for any given element z of 2A, there are 5 pairs (x,y) with x in 2A and y in 2B such that xy=z, and similarly for any given element z' in 2C there are 3 pairs (x,y) with x in 2A and y in 2B such that xy=z'.
Weyl groups of type An are created using CharacterTable("Symmetric",n+1), of type Bn=Cn using CharacterTable("WeylB",n), of type Dn using CharacterTable("WeylD",n), of type G2 is dihedral so CharacterTable("Dihedral",12), of type F4 using CharacterTable("W(F4)"), of type E6 using CharacterTable("W(E6)") etc.
